eAppendix 1: the SAS code for implementing the parametric bootstrap method.

```
data a;
	input a b y f;
	cards;
1 1 1 225
1 1 0 166
1 0 1   6
1 0 0  12
0 1 1   8
0 1 0  18
0 0 1   3
0 0 0  20
;
proc print data=a noobs;
	title "Example data";
	footnote "a & b are exposure factors, y is outcome, and f is 
frequency";
proc freq data=a noprint;
	weight f;
	tables a*b*y/sparse out=b;
proc sort data=b;
	by a b y;
data c;
	set b end=end;
	retain hold n00 p00 n01 p01 n10 p10 n11 p11;
	by a b y;
	if first.b then hold=count;
	if last.b then do;
		n=hold+count;
		p=count/n;
	end;
	if a=0 and b=0 then do; n00=n; p00=p; end;
	if a=1 and b=0 then do; n10=n; p10=p; end;
	if a=0 and b=1 then do; n01=n; p01=p; end;
	if a=1 and b=1 then do; n11=n; p11=p; end;
	if end then output;
	keep n00 p00 n01 p01 n10 p10 n11 p11;
data d;
	set c;
	call streaminit(3);
	do j=1 to 1000;
		y00=rand("binomial",p00,n00);
		y01=rand("binomial",p01,n01);
		y10=rand("binomial",p10,n10);
		y11=rand("binomial",p11,n11);
		hold=y11/(n11-y11+.5)-y10/(n10-y10+.5)-y01/(n01-y01+.5);
		reri=hold*(n00-y00)/(y00+.5)+1;
		output;
	end;
proc univariate data=d noprint;
	var reri;
	output out=e pctlpre=p pctlpts=2.5,97.5;
proc print data=e noobs round;
```

```
	title "95% CL for RERI from a nonparametric bootstrap with 
continuity correction";
	footnote "1000 resamples";
run;
quit;
run;
```

eAppendix 2: calculation of RERI with adjustment for confounders.

When there are no additional covariates, the purpose of the continuity correction used in equation (3) is to ensure $\exp\left(\hat{\beta}_1+\hat{\beta}_2+\hat{\beta}_3\right)$, $\exp\left(\hat{\beta}_1\right)$, and $\exp\left(\hat{\beta}_2\right)$ are bounded. For example, we replace $\exp\left(\hat{\beta}_1+\hat{\beta}_2+\hat{\beta}_3\right)=\frac{y_{11}}{n_{11}-y_{11}}\frac{n_{00}-y_{00}}{y_{00}}$ by $\frac{y_{11}}{n_{11}-y_{11}+.5}\frac{n_{00}-y_{00}}{y_{00}+.5}$ to ensure $\exp\left(\hat{\beta}_1+\hat{\beta}_2+\hat{\beta}_3\right)$ is bounded. With additional covariates the simple relation $\exp\left(\hat{\beta}_1+\hat{\beta}_2+\hat{\beta}_3\right)=\frac{y_{11}}{n_{11}-y_{11}}\frac{n_{00}-y_{00}}{y_{00}}$ does not hold, however we can directly constrain $\hat{\beta}_1,\hat{\beta}_2,\hat{\beta}_3$ so that $\exp\left(\hat{\beta}_1+\hat{\beta}_2+\hat{\beta}_3\right)$, $\exp\left(\hat{\beta}_1\right)$, and $\exp\left(\hat{\beta}_2\right)$ are bounded.

Given covariate *Z*=*z*, we assume

$$p_{ijz}=\frac{\exp\left(\beta_0+x_1\beta_1+x_2\beta_2+x_1x_2\beta_3+z\beta\right)}{1+\exp\left(\beta_0+x_1\beta_1+x_2\beta_2+x_1x_2\beta_3+z\beta\right)},$$

we fit the logistic regression model with constriant that the absolute values of $\hat{\beta}_1,\hat{\beta}_2,\hat{\beta}_3$ are below a threshold, and further compute RERI through equation (1), i.e.

$RERI=\exp\left(\hat{\beta}_1+\hat{\beta}_2+\hat{\beta}_3\right)-\exp\left(\hat{\beta}_1\right)-\exp\left(\hat{\beta}_2\right)+1$. For general consideration on fitting logistic regression models with constrained parameters, please see Tian and al (2008) [8] for details.